
Chương 4: CÁC ỨNG DỤNG DỰA TRÊN VI 
ĐIỀU KHIỂN MCS-51 

Chương này giới thiệu về một số ứng dụng của MCS-51 trong thực tế: điều 
khiển Led đơn, Led 7 đoạn, ma trận Led, LCD,  động cơ bước, giao tiếp 8255. 

1. Điều khiển Led đơn 
 

 

 

 

 

Hình 4.1 – Sơ đồ kết nối Led đơn 

Mạch điều khiển led đơn mô tả như hình 4.1. Lưu ý rằng các port của 
AT89C51 có dòng tối đa là 10 mA (xem thêm chương 1, phần đặc tính DC) nên khi 
cần điều khiển nhiều Led cần mắc thêm mạch khuếch đại. 

Hình 4.2 – Sơ đồ kết nối dùng mạch khuếch đại 

Hình 4.3 – Kết nối Led đơn với AT89C51 

Ví dụ: Xét sơ đồ kết nối Led như hình 4.3. Viết chương trình điều khiển Led 
sáng tuần tự từ trái sang phải, mỗi lần 1 Led. 

Giải 

Các Led nối với Port 0 của AT89C51 (P0 khi dùng như các cổng nhập / xuất 
thì cần phải có điện trở kéo lên nguồn) nên muốn Led sang thì phải gởi dữ liệu ra P0. 
Theo sơ đồ mạch, Led sang khi các bit tương ứng tại P0 là 0. 

Yêu cầu điều khiển Led sang từ trái sang phải (theo thứ tự lần lượt từ P0.0 đến 
P0.7) nên dữ liệu gởi ra là: 

- Lần 1: 1111 1110b (0FEh) – sáng 1 Led trái 

- Lần 2: 1111 1101b (0FDh) 

- Lần 3: 1111 1011b (0FBh) 

- Lần 4: 1111 0111b (0F7h) 

- Lần 5: 1110 1111b (0EFh) 

- Lần 6: 1101 1111b (0DFh) 

- Lần 7: 1011 1111b (0BFh) 

- Lần 8: 0111 1111b (7Fh) 

- Lần 9: quay lại giống như lần 1 

Chương trình thực hiện như sau: 

MOV DPTR,#MaLed  ; DPTR chứa vị trí bảng mã Led 
Main: 

MOV R7,#0    ; Phần tử đầu tiên của bảng mã 
Loop: 

MOV A,R7 
MOVC A,@A+DPTR  ; Đọc bảng mã 
MOV P0,A    ; Chuyển vào P0 để sáng Led 
CALL Delay   ; Chờ để mắt người có thể thấy 
INC R7    ; Chuyển qua trạng thái kế 
CJNE R7,#8,Loop  ; Đã hết bảng mã thì lặp lại 
SJMP main 

MaLed: DB 0FEh,0FDh,0FBh,0F7h,0EFh,0DFh,0BFh,7Fh 
Delay: 

MOV TMOD,#01h 
MOV TH0,#HIGH(-50000) ; Chờ 50 ms 
MOV TL0,#LOW(-50000) 
SETB TR0 
JNB TF0,$ 
CLR TF0 

CLR TR0 
RET 
END 

2. Điều khiển Led 7 đoạn 

2.1. Cấu trúc và bảng mã hiển thị dữ liệu trên Led 7 đoạn 

- Dạng Led: 
 

 

 

 

 

 

Hình 4.4 – Hình dạng của Led 7 đoạn 

 

- Led Anode chung: 
 

 

 

 

 

Hình 4.5 – Led 7 đoạn dạng anode chung 

Đối với dạng Led anode chung, chân COM phải có mức logic 1 và muốn sáng 
Led thì tương ứng các chân a – f, dp sẽ ở mức logic 0. 

Bảng 4.1 - Bảng mã cho Led Anode chung (a là MSB, dp là LSB): 

Số a b c d e f g dp Mã hex
0 0 0 0 0 0 0 1 1 03h 
1 1 0 0 1 1 1 1 1 9Fh 
2 0 0 1 0 0 1 0 1 25h 
3 0 0 0 0 1 1 0 1 0Dh 
4 1 0 0 1 1 0 0 1 99h 
5 0 1 0 0 1 0 0 1 49h 
6 0 1 0 0 0 0 0 1 41h 

7 0 0 0 1 1 1 1 1 1Fh 
8 0 0 0 0 0 0 0 1 01h 
9 0 0 0 0 1 0 0 1 09h 

Bảng 4.2 - Bảng mã cho Led Anode chung (a là LSB, dp là MSB): 

Số dp g f e d c b a Mã hex
0 1 1 0 0 0 0 0 0 0C0h 
1 1 1 1 1 1 0 0 1 0F9h 
2 1 0 1 0 0 1 0 0 0A4h 
3 1 0 1 1 0 0 0 0 0B0h 
4 1 0 0 1 1 0 0 1 99h 
5 1 0 0 1 0 0 1 0 92h 
6 1 0 0 0 0 0 1 0 82h 
7 1 1 1 1 1 0 0 0 0F8h 
8 1 0 0 0 0 0 0 0 80h 
9 1 0 0 1 0 0 0 0 90h 

 

- Led Cathode chung 
 

 

 

 

 

Hình 4.6 – Led 7 đoạn dạng cathode chung 

Đối với dạng Led Cathode chung, chân COM phải có mức logic 0 và muốn 
sáng Led thì tương ứng các chân a – f, dp sẽ ở mức logic 1. 

Bảng 4.3 - Bảng mã cho Led Cathode chung (a là MSB, dp là LSB): 

Số a b c d e f g dp Mã hex
0 1 1 1 1 1 1 0 0 0FCh 
1 0 1 1 0 0 0 0 0 60h 
2 1 1 0 1 1 0 1 0 0DAh 
3 1 1 1 1 0 0 1 0 0F2h 
4 0 1 1 0 0 1 1 0 66h 
5 1 0 1 1 0 1 1 0 0B6h 
6 1 0 1 1 1 1 1 0 0BEh 
7 1 1 1 0 0 0 0 0 0E0h 

8 1 1 1 1 1 1 1 0 0FEh 
9 1 1 1 1 0 1 1 0 0F6h 

 

Bảng 4.4 - Bảng mã cho Led Anode chung (a là LSB, dp là MSB): 

Số dp g f e d c b a Mã hex
0 0 0 1 1 1 1 1 1 3Fh 
1 0 0 0 0 0 1 1 0 06h 
2 0 1 0 1 1 0 1 1 5Bh 
3 0 1 0 0 1 1 1 1 4Fh 
4 0 1 1 0 0 1 1 0 66h 
5 0 1 1 0 1 1 0 1 6Dh 
6 0 1 1 1 1 1 0 1 7Dh 
7 0 0 0 0 0 1 1 1 07h 
8 0 1 1 1 1 1 1 1 7Fh 
9 0 1 1 0 1 1 1 1 6Fh 

 

2.2. Các phương pháp hiển thị dữ liệu 

2.2.1. Phương pháp quét 
Khi kết nối chung các đường dữ liệu của Led 7 đoạn, các Led không thể sáng 

đồng thời (do ảnh hưởng lẫn nhau giữa các Led) mà phải thực hiện quét Led, nghĩa là 
tại mỗi thời điểm chỉ sáng một Led và tắt các Led còn lại. Do hiện tượng lưu ảnh của 
mắt, ta sẽ thấy các Led sáng đồng thời. 

Ví dụ 1: Xét sơ đồ kết nối như hình 4.7. Viết chương trình hiển thị số 0 ra Led1 
và số 1 ra Led2. 

Giải 

Led có chân COM nối với Vcc (thông qua Q2, Q3) nên Led là loại anode 
chung và Q2, Q3 là transistor PNP nên để Led sáng thì dữ liệu tương ứng tại các chân 
điều khiển (P1.0, P1.1) phải là 1. 

Theo sơ đồ kết nối, chân g của Led nối với P0.6, chân a nối với P0.0 nên bảng 
mã Led là bảng 4.2, dữ liệu cho số 0 và 1 lần lượt là 0C0h và 0F9h. 

Phương pháp sử dụng là phương pháp quét nên cần phải có thời gian trì hoãn 
giữa 2 lần quét, thời gian này được thực hiện thông qua timer (thời gian trì hoãn 
khoảng 200 µs). 

Chương trình thực hiện như sau: 

MOV P1,#0  ; Xoá P1 để tắt Led 
Main: 

MOV P0,#0C0h ; Mã số 0 
SETB P1.0  ; Sáng Led1 
CALL Delay ; Thời gian trì hoãn để thấy Led sáng 
CLR P1.0  ; Tắt Led1 
MOV P0,#0F9h ; Mã số 1 
SETB P1.1  ; Sáng Led2 
CALL Delay 
CLR P1.1  ; Tắt Led2 
SJMP main 
;-------------------- 

Delay: 
MOV TMOD,#01h 
MOV TH0,#(-200) 
MOV TL0,#(-200) 
SETB TR0 
JNB TF0,$ 
CLR TF0 
CLR TR0 
RET 

END 

Ví dụ 2: Viết lại chương trình trên nhưng sử dụng ngắt của timer. 

Giải 

Đối với chương trình trong ví dụ 1, khi đang thực hiện quét led thì chương 
trình không làm gì cả trong khi đó, các ứng dụng thực tế thường xử lý các công việc 
khác đồng thời với quá trình quét. Vấn đề này có thể giải quyết bằng cách sử dụng 
ngắt của timer: mỗi khi timer tràn thì thực hiện hiển thị trên 1 Led. 

Chương trình thực hiện như sau: 

Led1 EQU 30h  ; Địa chỉ chứa dữ liệu của Led1 
Led2 EQU 31h  ; Địa chỉ chứa dữ liệu của Led2 
Led_Pos EQU 32h ; Vị trí Led hiện hành 
ORG 0000h 
LJMP main 
ORG 000Bh   ; Địa chỉ ISR của timer 0 
LJMP Timer0_ISR 

Main: 
SETB EA   ; Cho phép ngắt tại timer 0 
SETB ET0 
MOV Led1,#0C0h ; Số 0 
MOV Led2,#0F9h ; Số 1 
MOV Led_Pos,#01h ; Vị trí sáng đầu tiên là Led1 
MOV R0,#Led1 ; Dữ liệu gởi ra đầu tiên là ở Led1 
MOV TMOD,#01h 
MOV TH0,#(-200) 
MOV TL0,#(-200) 
SETB TR0 
SJMP $  ; Không làm gì cả, các ứng dụng thực tế 
   ; có thể thêm chương trình vào 
;---------------------- 

Timer0_ISR: 
MOV A,Led_Pos  ; Xác định vị trí Led hiện hành 
MOV P1,A   ; Sáng Led hiện hành 
RL A    ; Dịch trái để chuyển qua Led kế 
MOV Led_Pos,A  ; trong qua trình tràn tiếp theo 
MOV A,@R0   ; Đọc dữ liệu hiện hành 
MOV P0,A 
INC R0   ; Chuyển qua dữ liệu kế 
CJNE R0,#Led_Pos,exitTimer0 ; Nếu đã quét hết toàn bộ 
MOV Led_Pos,#01h ; Led thì bắt đầu lại từ Led1 
MOV R0,#Led1 

exitTimer0: 
RETI 
END 

Ví dụ 2 có thể mở rộng thêm cho 8 Led trong đó các bit điều khiển từ P1.0 đến 
P1.7 bằng cách khai báo thêm các ô nhớ cho các Led như sau: 

Led1 EQU 30h  ; Địa chỉ chứa dữ liệu của Led1 
Led2 EQU 31h  ; Địa chỉ chứa dữ liệu của Led2 
Led3 EQU 32h 
Led4 EQU 33h 
Led5 EQU 34h 
Led6 EQU 35h 
Led7 EQU 36h 
Led8 EQU 37h 
Led_Pos EQU 38h ; Vị trí Led hiện hành 

Ví dụ 3: Viết chương trình hiển thị nội dung trong ô nhớ 30h ra 2 Led trong đó 
Led1 chứa số hàng chục và Led2 chứa số hàng đơn vị (giả sử giá trị trong ô nhớ 30h 
tối đa là 99). 

Giải 

Để xuất nội dung trong ô nhớ 30h ra Led 7 đoạn cần thực hiện: 

- Chuyển nội dung trong ô nhớ 30h thành số hàng chục và hàng đơn vị (thực 
hiện chia cho 10). 

- Chuyển giá trị số thành mã Led 7 đoạn (bằng cách tra bảng). 

Chương trình thực hiện như sau: 

Led1 EQU 30h  ; Địa chỉ chứa dữ liệu của Led1 
Led2 EQU 31h  ; Địa chỉ chứa dữ liệu của Led2 
Led_Pos EQU 32h ; Vị trí Led hiện hành 
ORG 0000h 
LJMP main 
ORG 000Bh   ; Địa chỉ ISR của timer 0 
LJMP Timer0_ISR 

Main: 
SETB EA   ; Cho phép ngắt tại timer 0 
SETB ET0 
MOV Led_Pos,#01h ; Vị trí sáng đầu tiên là Led1 
MOV R0,#Led1 ; Dữ liệu gởi ra đầu tiên là ở Led1 
MOV TMOD,#01h 
MOV TH0,#(-200) 
MOV TL0,#(-200) 
SETB TR0 

Begin: 
MOV A,30h 
CALL Chuyenma  
SJMP Begin 
;---------------------- 

Chuyenma: 
MOV B,#10  ; Chia cho 10: A chứa số hàng chục, 
DIV  AB  ; B chứa số hàng đơn vị 
CALL BCDtoLed7 ; Chuyển sang mã Led 7 đoạn 
MOV Led1,A ; Đưa vào ô nhớ 31h (Led1) 
MOV A,B  ; Chuyển sang mã Led 7 đoạn của 
CALL BCDtoLed7; số hàng đơn vị 
MOV Led2,A 

RET 
;---------------------- 

BCDtoLed7: 
MOV DPTR,#MaLed7 
MOVC A,@A+DPTR 
RET 

MaLed7: DB 0C0h,0F9h,0A4h,0B0h,99h,92h,82h,0F8h,80h,90h 
;---------------------- 

Timer0_ISR: 
PUSH ACC 
MOV A,Led_Pos  ; Xác định vị trí Led hiện hành 
MOV P1,A   ; Sáng Led hiện hành 
RL A    ; Dịch trái để chuyển qua Led kế 
MOV Led_Pos,A  ; trong qua trình tràn tiếp theo 
MOV A,@R0   ; Đọc dữ liệu hiện hành 
MOV P0,A 
INC R0   ; Chuyển qua dữ liệu kế 
CJNE R0,#Led_Pos,exitTimer0 ; Nếu đã quét hết toàn bộ 
MOV Led_Pos,#01h ; Led thì bắt đầu lại từ Led1 
MOV R0,#Led1 

exitTimer0: 
POP ACC 
RETI 
END 

2.2.2. Phương pháp chốt 
Khi thực hiện tách riêng các đường dữ liệu của Led, ta có thể cho phép các Led 

sáng đồng thời mà sẽ không có hiện tượng ảnh hưởng giữa các Led. IC chốt cho phép 
lưu trữ dữ liệu cho các Led có thể sử dụng là 74LS373, 74LS374. Khi thực hiện bằng 
phương pháp chốt, khi nào cần xuất dữ liệu ra Led thì gởi dữ liệu và tạo xung để chốt. 

Ví dụ: Xét sơ đồ mạch kết nối như hình 4.8. Viết chương trình xuất số 2 ra 
Led3 và số 3 ra Led4. 

Giải 

Do Led3 nối với 74LS374 (U5) điều khiển bằng chân P1.0 nên để hiển thị trên 
Led3, cần phải: 

- Xuất dữ liệu ra P0. 
- Kích xung tại chân P1.0 để chốt dữ liệu 

Hình 4.7 – Kết nối Led 7 đoạn dùng phương pháp quét

Hình 4.8 – Kết nối Led 7 đoạn dùng phương pháp chốt 

Chương trình thực hiện như sau: 

MOV P0,#0B0h 
CLR P1.0 
SETB P1.0 
MOV P0,#99h 
CLR P1.1 
SETB P1.1 
END 

3. Điều khiển ma trận Led 
 Ma trận LED bao gồm nhiều LED cùng nằm trong một vỏ chia thành nhiều cột 

và hàng, mỗi giao điểm giữa hàng và cột có thể có 1 LED (ma trận LED một màu) hay 
nhiều LED (2 LED tại một vị trí tạo thành ma trận LED 3 màu). Để LED tại một vị trí 
nào đó sáng thì phải cấp hiệu điện thế dương giữa Anode và Cathode. Trên cơ sở cấu 
trúc như vậy, ta có thể mở rộng hàng và cột của ma trận LED để tạo thành các bảng 
quang báo. 

 

 

 

 

 

 

Hình 4.9 – Hình dạng ma trận Led 

Kết nối của ma trận Led có 2 cách: anode nối với hàng, cathode nối với cột hay 
ngược lại. Sơ đồ kết nối mô tả như hình 4.10. Theo cấu trúc kết nối như hình vẽ, 2 
Led trên 2 cột không thể sáng đồng thời. Xét sơ đồ kết nối như mạch hình b, một Led 
sáng khi tương ứng hàng của Led = 0 và cột = 1. 

Giả sử ta cần sáng Led đồng thời tại hàng 1, cột 1 và hàng 2, cột 2. Như vậy, ta 
phải có hàng 1 = 0, cột 1 = 1 (sáng Led tại hàng 1, cột 1) và hàng 2 = 0, cột 2 = 1 
(sáng Led tại hàng 2, cột 2). Từ đó, do hàng 1 = 0, cột 2 = 1 và hàng 2 = 0, cột 2 = 1 
nên ta cũng có các Led tại hàng 1, cột 2 và hàng 2, cột 1 cũng sáng. Nghĩa là, khi ta 
cho 2 Led tại hàng 1, cột 1 và hàng 2, cột 2 sáng đồng thời thì sẽ dẫn đến các Led tại 
hàng 1, cột 2 và hàng 2, cột 1 cũng sáng. 

Do đó, để thực hiện sáng một ký tự trên ma trận Led, ta phải dùng cơ chế quét, 
tại mỗi thời điểm chỉ sáng 1 cột, các cột còn lại tắt đi nhưng nếu cho thời gian quét đủ 
nhanh thì ta vẫn thấy giống như các cột sáng đồng thời. 

Hình 4.10 – Sơ đồ kết nối ma trận Led 

Dữ liệu cho số 0: 

 X X X  

X    X

X    X

X    X

X    X

X    X

 X X X  

 

Để sáng số 0 trên ma trận Led, ta thực hiện quá trình quét như sau: 

Lần 1: Hàng = 0100 0001b, cột = 0001 0000b 
Lần 2: Hàng = 0011 1110b, cột = 0000 1000b 
Lần 3: Hàng = 0011 1110b, cột = 0000 0100b 
Lần 4: Hàng = 0011 1110b, cột = 0000 0010b 
Lần 5: Hàng = 0100 0001b, cột = 0000 0001b 

Ví dụ: Xét sơ đồ kết nối ma trận Led như hình 4.11. Viết chương trình sáng số 
0 trên ma trận Led. 

Giải 

main: 
MOV R0,#0 

lap: 
MOV A,R0 
MOV DPTR,#cot 
MOVC A,@A+DPTR ; Xuất cột 
MOV P1,A 
 
MOV A,R0 
MOV DPTR,#hang 
MOVC A,@A+DPTR 
MOV P0,A   ; Xuất hàng 
CALL delay  ; Tạo thời gian trì hoãn để thấy 
INC R0   ; Chuyển sang cột kế 
CJNE R0,#5,lap ; Nếu quét đủ 5 cột thì lặp lại 



Giáo trình Vi điều khiển Các ứng dụng dựa trên vi điều khiển MCS-51 

Phạm Hùng Kim Khánh Trang 110 

SJMP main 
;-------------------- 
delay: 

MOV TMOD,#01h 
MOV TL0,#LOW(-500) 
MOV TH0,#HIGH(-500) 
SETB TR0 
JNB TF0,$ 
CLR TF0 
CLR TR0 
RET 

;-------------------- 
cot: DB 01h,02h,04h,08h,10h 
hang: DB 41h,3Eh,3Eh,3Eh,41h 
END 

Ví dụ 2: Viết chương trình cho chuỗi ‘KTCN’ di chuyển từ trái sang phải trên 
ma trận Led. 

Giải 

Giải thuật để Led di chuyển từ trái sang phải tham khảo thêm tại Tài liệu Thí 
nghiệm Vi xử lý – Bài 3 (ma trận Led và bàn phím) (download tại Website 
http://eed.hutech.edu.vn). 

main2: 
MOV R2,#0 

main1: 
MOV R1,#20 ; Một ký tự quét 20 lần 

main: 
MOV R0,#0 

lap: 
MOV A,R0 
MOV DPTR,#cot 
MOVC A,@A+DPTR 
MOV P1,A 
 
MOV A,R0 
ADD A,R2 
MOV DPTR,#hang 
MOVC A,@A+DPTR 
MOV P0,A 

CALL delay 
INC R0 
CJNE R0,#5,lap 
 
DJNZ R1,main 
INC R2 
CJNE R2,#31,main1 ; Nếu quét hết chuỗi thì lặp lại 
SJMP main2 
;----------- 

delay: 
MOV TMOD,#01h 
MOV TL0,#LOW(-500) 
MOV TH0,#HIGH(-500) 
SETB TR0 
JNB TF0,$ 
CLR TF0 
CLR TR0 
RET 

cot: DB 01h,02h,04h,08h,10h 
hang: DB 00h,77h,6Bh,5Dh,3Eh,7Fh ;Mã chữ K 

DB 7Eh,7Eh,00h,7Eh,7Eh,7Fh ;Mã chữ T 
DB 41h,3Eh,3Eh,3Eh,5Dh,7Fh ;Mã chữ C 
DB 00h,7Dh,7Bh,77h,00h,7Fh ;Mã chữ N 
DB 7Fh,7Fh,7Fh,7Fh,7Fh  ; Các cột trống 

END 

4. Điều khiển động cơ bước 
Động cơ bước là động cơ cho phép dịch chuyển mỗi lần một bước hay nửa 

bước tuỳ theo xung điều khiển. Góc quay của mỗi bước tuỳ theo loại động cơ, thường 
là 1.80/bước hay 7.20/bước. 

Động cơ bước gồm 4 cuộn dây: 1-2, 2-3, 4-5 và 5-6 như sơ đồ sau: 

 

 

 

 

Hình 4.12 – Động cơ bước 

Mạch điều khiển động cơ như sau: 

 

 

 

 

 

 

 

 

 

 

 

Hình 4.13 – Sơ đồ điều khiển động cơ bước 

Xung điều khiển động cơ như sau: 

Bảng 4.5 - Điều khiển một bước 
Ngược Thuận 

1 2 3 4 1 2 3 4 
1 0 0 0 1 0 0 0 
0 1 0 0 0 0 0 1 
0 0 1 0 0 0 1 0 
0 0 0 1 0 1 0 0 
1 0 0 0 1 0 0 0 

 

Bảng 4.6 - Điều khiển nửa bước 
Ngược Thuận 

1 2 3 4 1 2 3 4 
1 0 0 1 1 0 0 1 
1 0 0 0 0 0 0 1 
1 1 0 0 0 0 1 1 
0 1 0 0 0 0 1 0 
0 1 1 0 0 1 1 0 

0 0 1 0 0 1 0 0 
0 0 1 1 1 1 0 0 
0 0 0 1 1 0 0 0 
1 0 0 1 1 0 0 1 

 

Ví dụ: Xét sơ đồ kết nối động cơ như hình 4.14. Viết chương trình điều khiển 
động cơ quay thuận mỗi lần một bước với tốc độ 50 vòng/phút (giả sử động cơ có góc 
quay là 7.20/bước). 

1
2
3

4 5 6

MG1

MOTOR STEPPER

VCC

Q4

Q5

Q6

Q7

D13

D14

D15

D16

R11

R12

R13

R14

RST
9

XTAL2
18 XTAL1
19

PSEN
29ALE/PROG
30

EA/VPP
31

P1.0
1

P1.1
2

P1.2
3

P1.3
4

P1.4
5

P1.5
6

P1.6
7

P1.7
8

P2.0/A8
21

P2.1/A9
22

P2.2/A10
23

P2.3/A11
24

P2.4/A12
25

P2.5/A13
26

P2.6/A14
27

P2.7/A15
28

P3.0/RXD
10

P3.1/TXD
11

P3.2/INT0
12

P3.3/INT1
13

P3.4/T0
14

P3.5/T1
15

P3.6/WR
16

P3.7/RD
17

P0.0/AD0
39

P0.1/AD1
38

P0.2/AD2
37

P0.3/AD3
36

P0.4/AD4
35

P0.5/AD5
34

P0.6/AD6
33

P0.7/AD7
32

U8

AT89C51

 

Hình 4.14 – Sơ đồ kết nối AT89C51 với động cơ bước 

Giải 

Góc quay 7.20/bước → 1 vòng quay cần 3600/7.20 = 50 bước → 50 vòng quay 
cần thực hiện 2500 bước. 

Tốc độ 50 vòng / phút → 1 phút (60s) thực hiện 2500 bước → mỗi bước cần 
60/2500 = 0.024s = 24,,000 µs. 

Thứ tự kích xung như bảng 4.5. Chương trình thực hiện như sau: 

main: 
MOV R0,#0 

 MOV DPTR,#thuan1buoc 

begin: 
 MOV A,R0 
 MOVC A,@A+DPTR 

MOV P2,A   ; Xuất ra P2 để điều khiển động cơ 
CALL Delay 
INC R0 
CJNE R0,#4,begin 
SJMP main 
;----------------- 

Delay: 
MOV TMOD,#01h 
MOV TH0,#HIGH(-24000)  
MOV TL0,#LOW(-24000) 
SETB TR0 
JNB TF0,$ 
CLR TF0 
CLR TR0 
RET 
 

thuan1buoc: DB 08h,04h,02h,01h 
END 

 

5. Điều khiển LCD (Liquid Crystal Display) 
 Sơ đồ của LCD1602A: 

D
0

7

D
1

8

D
2

9

D
3

10

D
4

11

D
5

12

D
6

13

D
7

14

R
S

4

R
/W

5

E
N

6

C
O

N
S

T
3

G
N

D
1

V
C

C
2

A
15

K
16

THIS IS THE LCD
2 LINES x 16 CHARACTERS LA

MP

1602 LCD

 

Hình 4.15 – LCD 1602A 

- CONST (contrast): chỉnh độ tương phản (độ sáng của hình ảnh trên LCD). 
- EN (Enable): cho phép đọc/ghi dữ liệu. Trong chế độ đọc, EN tác động bằng 

xung dương (cạnh lên) và trong chế độ ghi, EN tác động bằng xung âm (cạnh 
xuống). 

- RS (register selection): chọn thanh ghi lệnh (RS = 0) hoặc thanh ghi dữ liệu 
(RS = 1) 

- R/W: đọc (R/W = 1) hay ghi (R/W = 0) 
- D7 – D4: bus dữ liệu (chế độ 8 bit: 4 bit cao, chế độ 4 bit: dùng cho truyền 4 

bit cao và 4 bit thấp). Ngoài ra, bit D7 còn dùng làm ngõ ra cho cờ Busy. 
- D3 – D0: 4 bit thấp trong chế độ 8 bit hay bỏ trống trong chế độ 4 bit. 
- A, K: anode và cathode đèn nền của LCD. 

 Các thành phần chức năng của LCD1602A: 

- Cờ Busy (BF – Busy flag): Nếu BF = 1, LCD đang trong quá trình thực thi 
một lệnh. Khi đó, các lệnh gởi tiếp theo sẽ bị bỏ qua. BF được đọc tại chân 
D7 khi RS = 0 và R/W = 1. Do đó, trước khi thực hiện một lệnh, cần kiểm tra 
BF trước, nếu BF = 0 thì mới gởi lệnh. 

- DDRAM (Display Data RAM): chứa các ký tự sẽ hiển thị trên LCD, tối đa 
là 80x8 bit (80 ký tự). Khi hiển thị ở chế độ 1 dòng, địa chỉ của DDRAM có 
phạm vi từ 00h ÷ 4Fh còn khi ở chế độ 2 dòng, địa chỉ DDRAM từ  00h ÷ 
27h cho dòng 1 và 40h ÷ 67h cho dòng 2. 

- Bộ đếm địa chỉ (AC - Address Counter): dùng để lưu địa chỉ hiện hành của 
DDRAM và CGRAM, có thể thực hiện đọc AC khi RS = 0 và R/W = 1. 

- CGROM (Character Genaration ROM): chứa các mô hình ký tự sẽ hiển 
thị trên LCD, bao gồm 192 ký tự 5x7 theo bảng mã ASCII (nghĩa là khi 
DDRAM chứa giá trị 41h tương ứng với mã ASCII của ký tự ‘A’ thì trên 
LCD sẽ hiện ‘A’), trong đó chỉ có các mã từ 00h – 0Fh sẽ không lấy theo mã 
ASCII mà lấy theo các ký tự đã định nghĩa trong CGRAM. 

- CGRAM (Character Genaration RAM): chứa các mô hình ký tự do người 
sử dụng định nghĩa để hiển thị các ký tự không có sẵn trong CGROM. 
CGRAM cho phép tạo tối đa 8 ký tự 5x8 (xem bảng 4.7). 

Bảng 4.7 – Các ký tự định nghĩa trong CGRAM 
DDRAM Địa chỉ CGRAM Dữ liệu CGRAM Ký tự 

00h hay 08h 000 000 
000 001 
000 010 
000 011 
000 100 
000 101 
000 110 
000 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

1 

01h hay 09h 001 000 
001 001 
001 010 
001 011 
001 100 
001 101 
001 110 
001 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

2 

02h hay 0Ah 010 000 
010 001 
010 010 
010 011 
010 100 
010 101 
010 110 
010 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

3 

03h hay 0Bh 011 000 
011 001 
011 010 
011 011 
011 100 
011 101 
011 110 
011 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

4 

04h hay 0Ch 100 000 
100 001 
100 010 
100 011 
100 100 
100 101 
100 110 
100 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

5 

05h hay 0Dh 101 000 
101 001 
101 010 
101 011 
101 100 
101 101 
101 110 
101 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

6 

06h hay 0Eh 110 000 
110 001 
110 010 
110 011 
110 100 
110 101 
110 110 
110 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

7 

07h hay 0Fh 111 000 
111 001 
111 010 
111 011 
111 100 
111 101 
111 110 
111 111 

xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 
xxx ????? 

8 

Để định nghĩa một ký tự, thực hiện thay thế dấu ? bằng các giá trị 0 hay 1 
tương ứng và gởi vào CGRAM. 

Ví dụ: Để định nghĩa chữ Đ tại vị trí 1 trong CGRAM, địa chỉ và dữ liệu tương 
ứng là: 

Địa chỉ Dữ liệu 

00h 1 1 1 1 0 1Eh 
01h 0 1 0 0 1 09h 
02h 0 1 0 0 1 09h 
03h 1 1 1 0 1 1Dh
04h 0 1 0 0 1 09h 
05h 0 1 0 0 1 09h 
06h 1 1 1 1 0 1Eh 
07h 0 0 0 0 0 00h 

 

Nghĩa là tại địa chỉ 00h của CGRAM chứa giá trị là 1Eh và tương tự cho đến 
địa chỉ 07h. 

 Các chế độ truyền dữ liệu: 

LCD1602A có 2 chế độ truyền dữ liệu: chế độ 8 bit (dùng cả D0 – D7) và chế 
độ 4 bit (không dùng D3 – D0, chỉ dùng D7 – D4). Trong trường hợp dùng chế độ 4 
bit, dữ liệu 8 bit sẽ được truyền 2 lần: truyền 4 bit cao rồi tiếp tục truyền 4 bit thấp. 

Sau khi thực hiện truyền xong 8 bit, BF mới chuyển lên 1. Hai chế độ truyền này mô 
tả như hình 4.16 và 4.17. 

 

Hình 4.16 – Định thời giao tiếp ở chế độ 8 bit 

 

Hình 4.17 – Định thời giao tiếp ở chế độ 4 bit 

 Tập lệnh: 

Bảng 4.8  - Tập lệnh của LCD1602A 

Lệnh 

Mã lệnh 

Mô tả 

Thời 
gian thực 
thi (fOSC 

= 270 
KHz) 

RS R/W D7 D6 D5 D4 D3 D2 D1 D0 

Xoá màn 
hình 

0 0 0 0 0 0 0 0 0 1 

Ghi 20h 
(khoảng trắng) 
vào DDRAM 
và đặt địa chỉ 
DDRAM là 
00h 

1.53ms 

Về đầu 
chuỗi 

0 0 0 0 0 0 0 0 1 X 

Đặt địa chỉ 
DDRAM là 
00h và trả con 
trỏ về vị trí đầu 
(nội dung 
DDRAM 
không đổi) 

1.53ms 

Định chế 
độ 0 0 0 0 0 0 0 1 I/D S 

- Chiều di 
chuyển con trỏ 
I/D = 1: tăng 
I/D = 0: giảm 
- Dịch toàn 
màn hình khi 
ghi DDRAM: 
S = 1: cho phép 
dịch 
S = 0: cấm 

39µs 

Điều 
khiển 
hiển thị 

0 0 0 0 0 0 1 D C B 

D = 1: hiện 
màn hình 
D = 0: cấm 
C = 1: hiện con 
trỏ 
C = 0: cấm 
B = 1: nhấp 
nháy 
B = 0: cấm 

39µs 

Dịch con 
trỏ hay 
màn hình 

0 0 0 0 0 1 S/C R/L X X 

S/C = 1: dịch 
màn hình 
S/C = 0: dịch 
con trỏ 
R/L = 1: dịch 
phải 
R/L = 0: dịch 
trái 

39µs 

Chức 
năng 

0 0 0 0 1 DL N F X X 

DL = 1: 8 bit 
DL = 0: 4 bit 
N = 1: 2 dòng 
N = 0: 1 dòng 

39µs 

F = 0: ký tự 
5x7 
F = 1: ký tự 
5x10 

Định địa 
chỉ 
CGRAM 

0 0 0 1 AC5 AC4 AC3 AC2 AC1 AC0 
Xác định địa 
chỉ của 
CGRAM 

39µs 

Định địa 
chỉ 
DDRAM 

0 0 1 AC6 AC5 AC4 AC3 AC2 AC1 AC0 
Xác định địa 
chỉ của 
DDRAM 

39µs 

Đọc BF 
và địa chỉ 
hiện hành 

0 1 BF AC6 AC5 AC4 AC3 AC2 AC1 AC0 

Xác định địa 
chỉ hiện hành 
và kiểm tra 
xem có thể gởi 
lệnh tiếp hay 
không 
BF = 1: không 
BF = 0: có thể 

0 

Ghi dữ 
liệu 1 0 D7 D6 D5 D4 D3 D2 D1 D0 

Ghi dữ liệu vào 
DDRAM hay 
CGRAM 

43µs 

Đọc dữ 
liệu 

1 1 D7 D6 D5 D4 D3 D2 D1 D0 
Đọc dữ liệu từ 
DDRAM hay 
CGRAM 

43µs 

I/D: Increment/Decrement  S: Screen  S/C: Screen/Cursor 
R/L: Right/Left   DL: Data Length N: Line number 
F: Font type    AC: Address Counter 

Các giá trị thường dùng mô tả như sau: 

Bảng 4.9 – Các lệnh thường dùng 
Lệnh Mô tả 

01H Xóa màn hình màn hình 
02H Trở về đầu chuỗi 
04H Dịch con trỏ sang trái  
06H Dịch con trỏ sang phải  
05H  Dịch màn hình sang phải  
07H Dịch màn hình sang trái  
08H Tắt con trỏ, tắt hiển thị 
0AH Tắt hiển thị, bật con trỏ  
0CH Bật hiển thị, tắt con trỏ  
0EH Bật hiển thị, nhấp nháy con trỏ  
0FH Tắt hiển thị, nhấp nháy con trỏ  
10H Dịch vị trí con trỏ sang trái  
14H  Dịch vị trí con trỏ sang phải  
18H Dịch toàn bộ màn hình sang trái  

1CH Dịch toàn bộ màn hình sang phải  
80H Đưa con trỏ về đầu dòng 1 
C0H Đưa con trỏ về đầu dòng 2 
38H Xác lập chế độ 2 dòng và độ phân giải chữ 5x7 

 

Ví dụ 1: Cho sơ đồ kết nối LCD 1602A với AT89C51 như hình vẽ. Viết 
chương trình hiển thị chuỗi “KHOA DIEN – DIEN TU” trên dòng 1 và “BO MON 
DIEN TU – VIEN THONG” trên dòng 2. 

Hình 4.18 – Kết nối LCD và 89C51 

Giải 

8 bit dữ liệu của LCD nối với P1 → chế độ 8 bit. Yêu cầu hiện trên 2 dòng → 
chế độ 2 dòng. 

Chương trình thực hiện như sau: 

EN   BIT   P2.2 
RS   BIT   P2.0 

RW   BIT   P2.1 
LCD_DATA    EQU      P1 
;---------------------------------------------------- 

main: 
MOV LCD_DATA,#38h  ; đặt chế độ 2 dòng 
CALL write_command 
 
MOV LCD_DATA,#0Ch  ; bật hiển thị  
CALL write_command 
 
MOV LCD_DATA,#01h    ;xoá màn hình  
CALL write_command 
 
MOV LCD_DATA,#80h  ; Chuyển về địa chỉ 00h (dòng 1) 
CALL write_command 
MOV DPTR,#Line1 
CALL write  ; Ghi vao DDRAM 
 
MOV LCD_DATA,#0C0h ; Chuyển về địa chỉ 40h (dòng 2) 
CALL write_command 
MOV DPTR,#Line2 
CALL write  ; Ghi vao DDRAM 
 
SJMP $ 

;-------------------------------------------------------- 
write: 

CLR A 
MOVC A,@A+DPTR 
CJNE A,#0FFh,write1;Nếu giá trị là 0FFh thì hết chuỗi 
RET 

write1: 
MOV LCD_DATA,A 
call write_data 
INC DPTR 
SJMP write 

;-------------------------------------------------- 
write_command: 

CLR RS  
CLR RW  
CLR EN 
NOP 

SETB EN 
NOP 
CLR EN 
CALL Delay 
RET 

;-------------------------------------------------- 
write_data: 

SETB RS 
CLR RW 
CLR EN  
NOP 
SETB EN 
NOP 
CLR EN 
CALL Delay 
RET 

;-------------------------------------------------- 
Delay: 

PUSH 07h 
PUSH 06h 
MOV R6,#50 
MOV R7,#255 
DJNZ R7,$ 
DJNZ R6,$-4 
POP 06h 
POP 07h 
RET 
;-------------------------------------------------- 
Line1: DB 'KHOA DIEN – DIEN TU', 0FFh 
Line2: DB 'BO MON DIEN TU – VIEN THONG', 0FFH 
END 

Ví dụ 2: Yêu cầu giống như ví dụ 1 nhưng cứ mỗi 1s thì dịch chuỗi sang trái 
một ký tự. 

Giải 

Chương trình thực hiện như trên nhưng thêm phần xử lý ngắt cho timer 0: cứ 
định thời 1s thì dịch chuỗi sang trái (nghĩa là dịch toàn bộ màn hình sang phải). Theo 
bảng 4.9, lệnh cần gởi ra LCD có mã lệnh là 1Ch. 

Chương trình thực hiện như sau: 

… 
ORG 0000h 
LJMP main 
ORG 000Bh 
LJMP Timer0_ISR 

Main: 
MOV IE,#82h ; Cho phép ngắt tại Timer 0 
MOV TMOD,#01h 
MOV TH0,#HIGH(-50000) 
MOV TL0,#LOW(-50000) 
MOV R7,#20 
SETB TR0 
… 

Timer0_ISR: 
MOV TH0,#HIGH(-50000) 
MOV TL0,#LOW(-50000) 
DJNZ R7,exitTimer0 
MOV R7,#20 
MOV LCD_DATA,#1Ch ;Dịch toàn màn hình sang phải 
CALL write_command 

exitTimer0: 
RETI 
… 
END 

Ví dụ 3: Cho mạch kết nối LCD như hình 4.18, viết chương trình xuất chuỗi 
“Khoa Điện – Điện tử” trên dòng 1 và “Bộ môn Điện tử - Viễn thông” trên dòng 2. 

Giải 

Ví dụ này yêu cầu các ký tự không có trong bảng mã nên phải định nghĩa thêm 
trong CGRAM. Các ký tự cần định nghĩa là: Đ, ệ, ử, ộ, ô, ễ, tổng cộng là 6 ký tự (có 
thể thực hiện được do LCD 1602A cho phép định nghĩa tối đa 8 ký tự). 

Địa chỉ và dữ liệu tương ứng là: 

Địa chỉ Dữ liệu Ký tự Mã DDRAM 
00h 1 1 1 1 0 1Eh 

Đ 00h 

01h 0 1 0 0 1 09h 
02h 0 1 0 0 1 09h 
03h 1 1 1 0 1 1Dh
04h 0 1 0 0 1 09h 
05h 0 1 0 0 1 09h 
06h 1 1 1 1 0 1Eh 
07h 0 0 0 0 0 00h 
08h 0 0 1 0 0 04h 

ệ 01h 

09h 0 1 1 1 0 0Eh 
0Ah 1 0 0 0 1 11h 
0Bh 1 1 1 1 0 1Eh 
0Ch 1 0 0 0 0 10h 
0Dh 0 1 1 1 1 0Fh 
0Eh 0 0 1 0 0 04h 
0Fh 0 0 1 0 0 00h 
10h 0 1 0 0 0 08h 

ử 02h 

11h 0 0 1 0 1 05h 
12h 0 1 0 0 1 09h 
13h 1 0 0 1 0 12h 
14h 1 0 0 1 0 12h 
15h 1 0 0 1 0 12h 
16h 0 1 1 0 0 0Ch
17h 0 0 0 0 0 00h 
18h 0 0 1 0 0 04h 

ộ 03h 

19h 0 1 0 1 0 0Ah
1Ah 0 1 1 1 0 0Eh 
1Bh 1 0 0 0 1 11h 
1Ch 1 0 0 0 1 11h 
1Dh 0 1 1 1 0 0Eh 
1Eh 0 0 1 0 0 04h 
1Fh 0 0 0 0 0 00h 

Địa chỉ Dữ liệu Ký tự Mã DDRAM 
20h 0 0 1 0 0 04h 

ô 04h 

21h 0 1 0 1 0 0Ah
22h 0 1 1 1 0 0Eh 
23h 1 0 0 0 1 11h 
24h 1 0 0 0 1 11h 
25h 0 1 1 1 0 0Eh 
26h 0 0 0 0 0 00h 
27h 0 0 0 0 0 00h 
28h   1  1 05h 

ễ 05h 

29h  1 1 1  0Eh 
2Ah  1  1  0Ah
2Bh 1 1 1 1 1 1Fh 
2Ch 1 1 1 1 1 1Fh 
2Dh 1     10h 
2Eh  1 1 1 1 0Fh 
2Fh 0 0 0 0 0 00h 

Chương trình thực hiện như sau: 

EN   BIT   P2.2 
RS   BIT   P2.0 
RW   BIT   P2.1 
LCD_DATA    EQU      P1 
;---------------------------------------------------- 
org 0 
ljmp main 
main: 
MOV LCD_DATA,#38h 
CALL write_command 
 
MOV LCD_DATA,#0Ch  
CALL write_command 
 
MOV LCD_DATA,#01h    ;xoá màn hình  
CALL write_command 
 
MOV LCD_DATA,#40h  ; Địa chỉ đầu của CGRAM 
call write_command  ; là 00h 
MOV DPTR,#cgram_data 
CALL write 

MOV LCD_DATA,#80h 
CALL write_command 
MOV DPTR,#Line1 
CALL write 
 
MOV LCD_DATA,#0C0h  
CALL write_command 
MOV DPTR,#Line2 
CALL write 
 
here:SJMP here 

;-------------------------------------------------------- 
write: 

CLR A 
MOVC A,@A+DPTR 
CJNE A,#0FFh,write1 
RET 

write1: 
MOV LCD_DATA,A 
call write_data 
INC DPTR 
SJMP write 

;-------------------------------------------------------- 
Delay: 

PUSH 07h 
PUSH 06h 
MOV R6,#50 
MOV R7,#255 
DJNZ R7,$ 
DJNZ R6,$-4 
POP 06h 
POP 07h 
RET 
;-------------------------------------------------- 

write_command: 
CLR RS  
CLR RW  
CLR EN 
NOP 
SETB EN 
NOP 

CLR EN 
SJMP wait 

;-------------------------------------------------- 
write_data: 

SETB RS 
CLR RW 
CLR EN  
NOP 
SETB EN 
NOP 
CLR EN 

wait: 
call delay 
ret 
 

;------------------------------------------------- 
Line1: DB 'Khoa ',00h,'i',01h,'n - ',00h,'i',01h,'n 
t',02h,0FFh ; Chuỗi ‘Khoa Điện – Điện tử’ 
Line2: DB 'B',03h,' m',04h,'n ',00h,'i',01h,'n t',02h,' - 
vi',05h,'n th',04h,'ng', 0FFH ; Chuỗi ‘Bộ môn Điện tử - 
Viễn thông’ 
;------------------------------------------------- 
cgram_data: DB 1Eh,09h,09h,1Dh,09h,09h,1Eh,00h ; Chữ Đ 

  DB 04h,0Eh,11h,1Eh,10h,0Fh,04h,00h ; Chữ ệ 
  DB 08h,05h,09h,12h,12h,12h,0Ch,00h ; Chữ ử 
  DB 04h,0Ah,0Eh,11h,11h,0Eh,04h,00h ; Chữ ộ 
  DB 04h,0Ah,0Eh,11h,11h,0Eh,00h,00h ; Chữ ô 
  DB 05h,0Eh,0Ah,1Fh,1Fh,10h,0Fh,00h ; Chữ ễ 
  DB 0FFh 
END 

6. Giao tiếp với PPI8255 
PPI8255 là IC giao tiếp lập trình được, cho phép mở rộng port trong trường 

hợp các port của 89C51 không đủ dùng. Các chế độ hoạt động của 8255 có thể tham 
khảo thêm tại Giáo trình Vi xử lý (cùng tác giả). 8255 có tổng cộng 2 chế độ: BSR 
(Bit Set/Reset) và I/O (Input/Output) trong đó I/O chia thành 3 chế độ khác nhau, tron 
tài liệu này chỉ xét ở chế độ 0 (xuất/nhập cơ bản). 

8255 có tổng cộng 3 port, mỗi port 8 bit trong đó port C có thể chia thành 4 bit 
cao và 4 bit thấp tạo thành 2 nhóm: nhóm A (PA + PCH) và nhóm B (PB và PCL). 

Hình 4.19 – Sơ đồ chân của 8255 

Để điều khiển 8255, bên trong có một thanh ghi điều khiển (CR – Control 
Register) cho phép chọn chế độ hoạt động. Nội dung của CR như sau: 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

D7 – D0: bus dữ liệu 

PA7 – PA0: Port A 

PB7 – PB0: Port B 

PC7 – PC0: Port C 

A1, A0: giải mã 

RESET: ngõ vào Reset 

CS : Chip Select 

RD : Read 

Hình 4.20 – Dạng từ điều khiển cho 8255A ở chế độ I/O 

D7 D6 D5 D4 D3 D2 D1 D0 

0 x x X    S/R 

Mode BSR Không sử dụng Chọn bit 
000: PC0 
001: PC1 
010: PC2 
011: PC3 
100: PC4 
101: PC5 
110: PC6 
111: PC7 

0: Xoá (Reset) 
1: Đặt (Set) 

Hình 4.21 - Dạng từ điều khiển cho 8255A ở chế độ BSR 

Lưu ý rằng khi cần Set/Reset bit thì phải gởi dữ liệu ra CR chứ không gởi ra 
PC. 

Như vậy, để xác lập điều khiện làm việc cho 8255, cần thực hiện định cấu hình 
cho 8255 (chọn các chế độ hoạt động cho PA, PB và PC). Để thực hiện quá trình này, 
cần tác động đến CR của 8255. Logic chọn các port cho 8255 mô tả như sau: 

Bảng 4.10 – Logic chọn các port của 8255 

CS  A1 A0 Chọn 

0 
0 
0 
0 
1 

0 
0 
1 
1 
x 

0 
1 
0 
1 
x 

Port A 
Port B 
Port C 
Thanh ghi điều khiển 
8255A không hoạt động 

 

Ví dụ: Cho mạch kết nối giữa AT89C51 và 8255 như hình 4.22. Viết chương 
trình điều khiển theo yêu cầu: 

- Nhấn SW1: sáng 4 Led trái và sáng Lamp. 

- Nhấn SW2: sáng 4 Led phải và tắt Lamp. 

Giải 

Do PA điều khiển Led, PCL điều khiển công tắc nhấn, PCH điều khiển RL1 
nên PA xuất, PCL nhập và PCH xuất (còn PB tuỳ ý). Nội dung thanh ghi điều khiển 
như sau: 

1 0 0 0 0 0 0 1 81h 
I/O Chế độ 0 PA xuất PCH xuất Chế độ 0 PB xuất PCL nhập  

 

Led đơn nối với các bit của PA tại cathode và anode nối với Vcc nên để Led 
sáng thì dữ liệu tại PA là 0 và Led tắt khi dữ liệu là 1. 

Đèn LAMP được điều khiển bằng RL1: khi RL1 đóng (ứng với PC7 = 0) thì 
LAMP sáng và ngược lại, khi RL1 ngắt (ứng với PC7 = 1) thì LAMP tắt. Nội dung 
thanh ghi điều khiển khi điều khiển PC7 như sau: 

PC7 = 0 
0 0 0 0 1 1 1 0 0Eh

BSR Không dùng PC7 = 0  
PC7 = 1 

0 0 0 0 1 1 1 1 0Fh
BSR Không dùng PC7 = 1  

 

Công tắc SW1, SW2 nối với PC0 và PC1: khi nhấn công tắc thì chân tương 
ứng tại PC = 0 và khi không nhấn thì = 1. Do đó, để kiểm tra công tắc có nhấn hay 
không thì đọc dữ liệu từ PCL và kiểm tra tương ứng các bit PC0, PC1. 

Địa chỉ các port của 8255: 

CS  = 0 
(Y6 = 0) 

Tuỳ ý A1 A0 Port Địa chỉ 
hex 

A15 A14 A13 A12 A11 A10 A9 A8 A7 A6 A5 A4 A3 A2 A1 A0   

1 1 0 X X X X X X X X X X X 0 
0 
1 
1 

0 
1 
0 
1 

A 
B 
C 
CR 

C000h 
C001h 
C002h 
C003h 

 

Chương trình thực hiện như sau: 

MOV DPTR,#0C003h ; Địa chỉ CR 
MOV A,#81h; PA: xuất, PB: xuất, PCH: xuất, PCL: nhập 
MOVX @DPTR,A  ; Xuất ra CR 

Begin: 
MOV DPTR,#0C002h ; Địa chỉ PC 

MOVX A,@DPTR  ; Đọc vào 
JNB ACC.0,SW1  ; Nếu PC0 = 0 thì đến SW1 
JNB ACC.1,SW2  ; Nếu PC1 = 0 thì đến SW2 
SJMP begin 

SW1: 
CALL Delay  ; Tránh rung phím 
MOV A,11110000b ; Sáng 4 Led trái 
MOV DPTR,#0C000h ; Địa chỉ PA (do PA nối với Led) 
MOVX @DPTR,A 
 

MOV A,0Eh   ; PC7 = 0 → đóng RL1 → sáng LAMP 
MOV DPTR,#0C003h ; Địa chỉ CR (do dùng chế độ BSR) 
MOVX @DPTR,A 
SJMP begin 

SW2: 
 CALL Delay 

MOV A,00001111b ; Sáng 4 Led phải 
MOV DPTR,#0C000h ; Địa chỉ PA (do PA nối với Led) 
MOVX @DPTR,A 
 

MOV A,0Fh   ; PC7 = 1 → đóng RL1 → sáng LAMP 
MOV DPTR,#0C003h ; Địa chỉ CR (do dùng chế độ BSR) 
MOVX @DPTR,A 
SJMP begin 

;---------------------------- 
Delay: 

MOV TMOD,#02h 
MOV TH0,#HIGH(-50000) 
MOV TL0,#LOW(-50000) 
SETB TR0 
JNB TF0,$ 
CLR TF0 
CLR TR0 
RET 
END 

BÀI TẬP CHƯƠNG 4 
1. Cho sơ đồ kết nối như hình 4.3. Viết chương trình sáng Led theo yêu cầu: 

sáng lần lượt 1 Led từ phải sang trái và thực hiện 4 lần; nhấp nháy 8 Led 5 
lần; sáng Led từ ngoài vào trong, mỗi lần 2 Led và thực hiện 3 lần (thời 
gian trì hoãn giữa 2 lần sáng là 300ms, dùng timer 1). 

2. Cho sơ đồ kết nối như hình 4.7. Viết chương trình tăng nội dung của ô nhớ 
30h từ 00 – 99 và hiển thị giá trị trên 2 Led 7 đoạn (hiển thị Led bằng ngắt 
timer 1 và thời gian trì hoãn khi tăng nội dung của ô nhớ 30h là 1s dùng 
ngắt timer 0). 

3. Cho sơ đồ kết nối như hình 4.7 trong đó kết nối thêm 4 Led (Led2 – 6) được 
điều khiển bằng các bit của P1: P1.2 – P1.5. Viết chương trình hiển thị giờ, 
phút giây trên 6 Led (Led1,2: giờ; Led3,4: phút; Led5,6: giây) trong đó giờ 
chứa trong ô nhớ 30h, phút trong ô nhớ 31h, giây trong ô nhớ 32h (thời gian 
trì hoãn 1s dùng ngắt timer 0, quét Led dùng ngắt timer 1). 

4. Cho sơ đồ kết nối như hình 4.11. Viết chương trình cho chuỗi “DAI HOC 
KY THUAT CONG NGHE TPHCM” di chuyển từ trái sang phải trên ma 
trận Led. 

5. Cho sơ đồ kết nối như hình 4.14. Viết chương trình điều khiển động cơ 
quay thuận 100 vòng với tốc độ 10 vòng/phút (giả sử mỗi bước có góc quay 
là 7.20). 

6. Cho sơ đồ kết nối như hình 4.18. Viết chương trình cho chuỗi “Đại học Kỹ 
thuật Công nghệ” trên dòng 1 và “Khoa Điện – Điện tử” trên dòng 2 di 
chuyển từ trái sang phải (thời gian dịch chuyển là 300ms dùng ngắt timer 
1). 

 


